*** Mader, Matthias. 2016. Stabilität und Wandel der nationalen Identität in der deutschen Bevölkerung
*** In: Kölner Zeitschrift für Soziologie und Sozialpsychologie. DOI: 10.1007/s11577-016-0370-9.

*** Reproduktionssyntax: Datenaubereitung, Mplus-Datenexport, Reproduktion Tabelle 1

*** Datendownload von GESIS: ZA2880 (Version 1), ZA3910 (Version 2), ZA5950  (Version 2)

*** Installation Stata Ados (letzter Zugriff: 19.7.2016)
	*ssc install fre
	*net install stata2mplus, from(http://www.ats.ucla.edu/stat/stata/ado/analysis)
	*net install runmplus, from(http://fmwww.bc.edu/RePEc/bocode/r)

	clear
	set more off
	cd "C:\Users\m.mader\Dropbox\Publikationen\Mader_2016_Nationale_Identitäten_in_Deutschland\Daten_und_Dokumentation_neu"
	
**************************************************************************************************************************************************************************************************************		
*** Datensatz: ISSP1995
*** Gesis-Studiennummer: ZA2880, Version 1

use ZA2880.dta, clear

*** Befragte aus anderen ISSP-Ländern löschen
	fre v3
	keep if v3 == 2 | v3 == 3
	
*** Nationale Identität

	*Patriotischer Nationalstolz
		fre v28 v31 v37, mis
		*Stolz Demokratie
		recode v28 (4 = 1) (3 = 2) (2 = 3) (1 = 4), gen(p1)
		*Stolz Gleichbehandlung
		recode v31 (4 = 1) (3 = 2) (2 = 3) (1 = 4), gen(p2)
		*Stolz Sozialleistungen 
		recode v37 (4 = 1) (3 = 2) (2 = 3) (1 = 4), gen(p3)
		
	*Chauvinismus
		fre v24 v25 v22
		*Bessere Welt
		mvdecode v24, mv(8 9 = .)
		recode v24 (5 = 1) (4 = 2) (2 = 4) (1 = 5), gen(chauv1)
		*D besser als andere
		mvdecode v25, mv(8 9 = .)
		recode v25 (5 = 1) (4 = 2) (2 = 4) (1 = 5), gen(chauv2)
	
	*völkisch-kulturalistische Haltungen
		fre v15 v44 v43 v36
		*Wichtigkeit, in D geboren zu sein
		recode v15 (4 = 1) (3 = 2) (2 = 3) (1 = 4), gen(vk1)	
		*Sitten und Gebräuche teilen
		mvdecode v44, mv(8 9 = .)	
		recode v44 (5 = 1) (4 = 2) (2 = 4) (1 = 5), gen(vk2)
		*Deutschen Filmen Vorzug geben
		fre v43
		recode v43 (5 = 1) (4 = 2) (2 = 4) (1 = 5), gen(vk3)
		
*** Ost-West-Dummy (Westberlin = Ostdeutschland)
	recode v324 (1 2 3 4 5 6 7 8 9 10 = 0) (11 12 13 14 15 16 = 1), gen(ost)
	tab v324 ost

*** Ost-West-Gewicht
	*siehe http://www.gesis.org/issp/modules/issp-modules-by-topic/national-identity/1995/
	gen owgewicht0 = .
	replace owgewicht0 = 1.19455 if ost == 0
	replace owgewicht0 = 0.59238 if ost == 1
	sum owgewicht0
	egen owgewicht0m = mean(owgewicht0)
	gen owgewicht = owgewicht0 / owgewicht0m
	
	*ID Variable Gewicht
	gen id_gewicht = 1
	replace id_gewicht = 0 if owgewicht == .
		
*ID Erhebungsjahr
	gen year = 1995

	*ID Befragte
	gen id = v2
			
*ISSP1995-Daten auf zwei Ausprägungen reduzieren
	foreach var of varlist chauv1 chauv2 vk2 vk3 {
		recode `var' (1 2 3 = 1) (4 5 = 2), gen(`var'x)
		}
	foreach var of varlist p1 p2 p3 vk1 {
		recode `var' (1 2 = 1) (3 4  = 2), gen(`var'x)
		}	
 
		 
*** Mplus-Export der ISSP1995-Daten
	stata2mplus p1 p2 p3 chauv1 chauv2 vk1 vk2 vk3 owgewicht using DATA_issp_1995 , replace		

	stata2mplus p1x p2x p3x chauv1x chauv2x vk1x vk2x vk3x owgewicht using DATA_issp_1995x, replace	

*** Datensatz auf relevante Variablen reduzieren und speichern
	keep p1 p2 p3 chauv1 chauv2 vk1 vk2 vk3 ///
		 p1x p2x p3x chauv1x chauv2x vk1x vk2x vk3x owgewicht year id id_gewicht
	save ISSP1995.dta, replace
			

			
***********************************************************************************************************************************************************************************************************************************	
*** Datensatz: ISSP2003
*** Gesis-Studiennummer: ZA3910, Version 2

	use ZA3910_v2-1-0.dta, clear

*** Befragte aus anderen ISSP-Ländern löschen
	fre COUNTRY
	keep if COUNTRY == 2 | COUNTRY == 3

	
*Nationale Identität

	*Patriotischer Nationalstolz
		fre v26 v29
		*Stolz Demokr
		fre v26
		recode v26 (4 = 1) (3 = 2) (2 = 3) (1 = 4), gen(p1)
		*Stolz Gleichbehandlung
		fre v35
		recode v35 (4 = 1) (3 = 2) (2 = 3) (1 = 4), gen(p2)
		*Stolz Sozialleistungen 
		fre v29
		recode v29 (4 = 1) (3 = 2) (2 = 3) (1 = 4), gen(p3)
	
	*völkisch-kulturalistische Haltungen
		fre v11 v47 v40
		*Wichtigkeit, in D geboren zu sein
		fre v11
		recode v11 (4 = 1) (3 = 2) (2 = 3) (1 = 4), gen(vk1)	
		*Sitten und Gebräuche teilen
		fre v47	
		recode v47 (5 = 1) (4 = 2) (2 = 4) (1 = 5), gen(vk2)
		*Deutschen Filmen Vorzug geben
		fre v40			
		recode v40 (5 = 1) (4 = 2) (2 = 4) (1 = 5), gen(vk3)
		
	*Chauvinismus
		fre  v21 v22
		*Bessere Welt
		fre v21
		recode v21 (5 = 1) (4 = 2) (2 = 4) (1 = 5), gen(chauv1)
		*D besser als andere
		fre v22
		recode v22 (5 = 1) (4 = 2) (2 = 4) (1 = 5), gen(chauv2)
		
*** Ost-West-Dummy		
	fre de_reg
	recode de_reg (1 2 3 4 5 6 7 8 9 10 17 = 0) (11 12 13 14 15 16 = 1), gen(ost)
	tab de_reg ost
		
*** Ost-West-Gewicht 2004
	*siehe http://www.gesis.org/issp/modules/issp-modules-by-topic/national-identity/2003/	
	gen owgewicht0 = .
	replace owgewicht0 = 1.232524 if ost == 0
	replace owgewicht0 = 0.547722 if ost == 1
	sum owgewicht0
	egen owgewicht0m = mean(owgewicht0)
	gen owgewicht = owgewicht0 / owgewicht0m
	sum owgewicht
	
	*ID Gewicht
	gen id_gewicht = 1
	replace id_gewicht = 0 if owgewicht == .
	
*** ID Erhebungsjahr
	gen year = 2004

*** ID Befragte
	gen id = V3

*** ISSP2003-Daten auf zwei Ausprägungen reduzieren und nach Mplus exportieren			
	foreach var of varlist chauv1 chauv2 vk2 vk3 {
		recode `var' (1 2 3 = 1) (4 5 = 2), gen(`var'x)
		}
	foreach var of varlist p1 p2 p3 vk1 {
		recode `var' (1 2 = 1) (3 4  = 2), gen(`var'x)
		}	
		
*** ISSP2003-Daten nach Mplus exportieren (Erhebungsjahr Deutschland: 2004)
	stata2mplus p1 p2 p3 chauv1 chauv2 vk1 vk2 vk3 owgewicht using DATA_issp_2004, replace	
					
	stata2mplus p1x p2x p3x chauv1x chauv2x vk1x vk2x vk3x owgewicht using DATA_issp_2004x, replace	
			
*** Datensatz auf relevante Variablen reduzieren und abspeichern
	keep p1 p2 p3 chauv1 chauv2 vk1 vk2 vk3 ///
		p1x p2x p3x chauv1x chauv2x vk1x vk2x vk3x ///
		owgewicht year id id_gewicht

	save ISSP2004.dta, replace
	
		
**************************************************************************************************************************************************************************************************************		
*** Datensatz: ISSP2013
*** Gesis-Studiennummer: ZA5950, Version 2

	use ZA5950_v2-0-0.dta, clear

*** Befragte aus anderen ISSP-Ländern löschen
	keep if  V4 == 276
	
*** Nationale Identität

	*Patriotischer Nationalstolz
		fre V25 V34 V28
		*Stolz Demokratie
		fre V25
		mvdecode V25, mv(8 9 = .)
		recode V25 (4 = 1) (3 = 2) (2 = 3) (1 = 4), gen(p1)
		*Stolz Gleichbehandlung
		fre V34
		mvdecode V34, mv(8 9 = .)
		recode V34 (4 = 1) (3 = 2) (2 = 3) (1 = 4), gen(p2)
		*Stolz Sozialleistungen 
		fre V28
		mvdecode V28, mv(8 9 = .)
		recode V28 (4 = 1) (3 = 2) (2 = 3) (1 = 4), gen(p3)
		
	*Chauvinismus
		fre V19 V20
		*Bessere Welt
		fre V19
		mvdecode V19, mv(8 9 = .)
		recode V19 (5 = 1) (4 = 2) (2 = 4) (1 = 5), gen(chauv1)
		*D besser als andere
		fre V20
		mvdecode V20, mv(8 9 = .)
		recode V20 (5 = 1) (4 = 2) (2 = 4) (1 = 5), gen(chauv2)
	
	*völkisch-kulturalistische Haltungen
		fre V9 V45 V39 
		*Wichtigkeit, in D geboren zu sein
		fre V9
		mvdecode V9, mv(8 9 = .)
		recode V9 (4 = 1) (3 = 2) (2 = 3) (1 = 4), gen(vk1)	
		*Sitten und Gebräuche teilen
		fre V45
		mvdecode V45, mv(8 9 = .)	
		recode V45 (5 = 1) (4 = 2) (2 = 4) (1 = 5), gen(vk2)
		*Deutschen Filmen Vorzug geben
		fre V39
		mvdecode V39, mv(8 9 = .)		
		recode V39 (5 = 1) (4 = 2) (2 = 4) (1 = 5), gen(vk3)
		
*** Ost-West-Gewicht 2014
	gen owgewicht = WEIGHT
	sum owgewicht
	
	*ID Variable Gewicht
	gen id_gewicht = 1
	replace id_gewicht = 0 if owgewicht == .
	
*** ID Befragte
	gen id = CASEID
	
*** ID Erhebungsjahr
	gen year = 2014
			
*** ISSP2013-Daten nach Mplus exportieren (Erhebungsjahr Deutschland: 2014)
	foreach var of varlist chauv1 chauv2 vk2 vk3 {
		recode `var' (1 2 3 = 1) (4 5 = 2), gen(`var'x)
		}
	foreach var of varlist p1 p2 p3 vk1 {
		recode `var' (1 2 = 1) (3 4  = 2), gen(`var'x)
		}	

*** ISSP2013-Daten nach Mplus exportieren (Erhebungsjahr Deutschland: 2014)
	stata2mplus p1 p2 p3 chauv1 chauv2 vk1 vk2 vk3 owgewicht using DATA_issp_2014, replace

	stata2mplus p1x p2x p3x chauv1x chauv2x vk1x vk2x vk3x owgewicht using DATA_issp_2014x, replace

*** Datensatz reduzieren
	keep p1 p2 p3 chauv1 chauv2 vk1 vk2 vk3 ///
		 p1x p2x p3x chauv1x chauv2x vk1x vk2x vk3x ///
		 year owgewicht id_gewicht id


*** Kumulation von ISSP1995, ISSP2003 und ISSP2013
	append using ISSP1995.dta
	append using ISSP2004.dta

	save DATA_STATA_issp.dta, replace
	erase ISSP2004.dta
	erase ISSP1995.dta
	
*** ISSP1995-, ISSP2003-, ISSP2013-Daten nach Mplus exportieren
	stata2mplus p1 p2 p3 chauv1 chauv2 vk1 vk2 vk3 ///
				year owgewicht ///
				using DATA_issp_1995_2004_2014, replace	
		

**************************************************************************************************************************************************************************************************************		
		
*** Analysen			

*** Tabelle 1: Randverteilungen der dichotomisierten Indikatoren
	foreach var of varlist p1x p2x p3x vk1x vk2x vk3x chauv1x chauv2x  {
		recode `var' (2 = 1) (1 = 0), gen(`var'x)
		}

	bysort year: sum p1xx p2xx p3xx vk1xx vk2xx vk3xx chauv1xx chauv2xx [aw = owgewicht]

*** Tabelle 1: Alphas
	bysort year: alpha p1 p2 p3
	bysort year: alpha vk1 vk2 vk3
	bysort year: alpha chauv1 chauv2
